

如何使用 HT46R7xD-1 双斜率积分器功能完成 AD 转换

文件编码: HA0118S

简介

本范例用来实现 HT46R7xD-1 中 Dual Slope ADC 的双斜率积分器功能完成 A/D 转换。利用 Timer 设定充电时间，充电完成后，同样利用 Timer 对放电时间计时，放电完成后，Timer 停止计数，Timer 的计数值就是仿真量对应的数字量。

功能与特性说明

双积分 A/D 主要由 4 个运算放大器组成(双积分 A/D 为 HT46R7xD-1 的内部组成)。从左到右依次为 Amplifier、Buffer、Integrator 和 Comparator。

应用电路图

IC 外围电路原理图




图 1 IC 外围电路原理图

Dual Slope ADC 的内部结构示意图




图 2 Dual Slope ADC 的内部结构示意图

H/W Operating Principle

Amplifier

Amplifier

用来放大 Sensor 的差分信号的。传感器出来信号是相当微弱的，电压一般都是毫伏级(mV)甚至微伏级(μV)，而后面的积分器需要有几百毫伏的电压，所以必须把传感器信号放大以供积分器使用。

Amplifier 的放大倍数是由硬件决定的，下图是输入放大器电路的接法：




图3 放大器接法

输入量 V_A 、 V_B 分别加到运放的同相端 (DOPAP) 和反相端 (DOPAN)，输出电压 V_{DOPA0} 则通过电阻 R_1 反馈到反相端，构成负反馈。

放大倍数由电阻 R_1 和 sensor 的等效电阻决定，即

$$V_{DOPA0} = (V_A - V_B) * R_1 / R_{\text{sensor}} \quad (\text{式 } 1)$$

同时 sensor 输出的差分信号 V_A 、 V_B 各自的对地电压最低不能低于 $0.2V$ ，最高不能高于 $V_{DD}-1V$ ，即 $0.2V < V_A(V_B) < V_{DD}-1V$ 。

Buffer

Buffer 这个运算放大器接成电压跟随器，在电路中起到阻抗匹配的作用，使得前面差动放大器出来的电压稳定，使后面的积分器不受到前面放大器的影响。

Integrator 和 Comparator

Integrator 也就是积分器，是双积分 A/D 中的关键部位，它和比较器一起构成 A/D 最关键的部分，如下图：




图 4 积分器和比较器

先介绍一下跟随器输出接的那个 MUX，是一个多任务器，它可以看做是一个开关，控制积分器是充电还是放电，由 A/D 的控制缓存器 ADCR 其中的两个位来实现。

- 充电过程

由图 4 可以看出，积分器的输出由电容 CDS 引回运算放大器反相端，够成负反馈。在引入负反馈的运算电路中，有两个重要的概念：虚短路和虚断路。由虚短可知，Integrator 的反相端的电压与同相端一样，也是 $4/6 VDSO$ ，即 $VDSRC = 4/6 VDSO$ ，当 MUX 切到 Buffer 端时，DSRR 脚的电压就是放大器放大后的电压，由于 SENSOR 的信号很微弱，放大后的电压也决不会高于 $4/6 VDSO$ ，即有 $VDSRR < VDSRC$ ，这样在电阻 RDS 两端就有了压差，导致 RDS 中有电流流过，图中红色箭头表示充电时电流的流向；再由虚断路，可得流入两输入端的电流等于零，电流只能从电容 CDS 流向电阻 RDS；又因电流的方向其实是正电荷移动的方向，所以正电荷就在 CDS 的右端慢慢积累，电压也就随之慢慢抬高，即 $VDSRC$ 慢慢抬高；当 $VDSRC > 1/6 VDSO$ 时，比较器就会输出高，表示充电开始。

- 放电过程

当设定的充电时间到时，把 MUX 切到 VDSO，此时 $VDSRR > VDSRC$ ，同样 RDS 有电流流过，电流方向如图 4 蓝色箭头所示。同样由于虚断路，流入运算放大器两输入端的电流等于零，电流只能从电阻 RDS 流向电容 CDS，电容右端正电荷随之减少，电压下降；当 $VDSRC < 1/6 VDSO$ 时，比较器输出低，表示放电结束，此时应把 MUX 断开，即停止 A/D 继续工作，防止 $VDSRC$ 继续往下降，为下一次的充电做好准备。

需要注意的是，积分器充、放电有两个参数是固定的：

- 充电时间是固定的
- 放电的斜率是固定的




图5 积分器充、放电图形

另外，使用 A/D 的时候，同时要打开电压调整器 (Regulator)，以保证 A/D 有稳定的工作电压。电压泵打开与否视电源电压是否大于 3.6V 而定，若大于 3.6V 则不必打开。

Dual Slope ADC 相关控制寄存器

以 HT46R71D-1 为例。

下面两个寄存器 ADCR 和 ADCD 用来控制双积分 A/D 的工作，再配合 Timer 的控制寄存器就可以完成对 A/D 的操作。


位	符号	功能
0	ADPWREN	双积分电路电源开关 0: 电源关闭 1: 电源来自稳压器
1	ADDISCH0	设置 AD 转换器充电/放电 (ADDISCH1:0) 00: 保留 01: 充电 (积分器输入连接至缓冲器输出) 10: 放电 (积分器输入连接至 VDSO) 11: 保留
2	ADDISCH1	
3	ADCMPO	双积分 AD 转换器---比较器输出 只读，写此位指令被忽略 在放电阶段，当积分器输出电压低于参考电压，ADCMPO 将由高变低
4~5	—	保留
6	ADCCKEN	AD 转换器断路器时钟开关 0: 关闭 1: 打开 (时钟频率由 ADCD 寄存器设置)
7	ADRR0	AD 转换器电阻选择 0: (V_{INT}, V_{CMP}) = (4/6VOREG, 1/6VOREG) 1: (V_{INT}, V_{CMP}) = (4.4/6VOREG, 1/6VOREG)

ADCR 控制寄存器

位	符号	功能	
0	ADCD0	定义断路器时钟 (需 ADCCKEN 使能)，建议频率为 10kHz 其断路器频率： 0: 时钟频率= ($f_{SYS}/32$) /1 1: 时钟频率= ($f_{SYS}/32$) /2 2: 时钟频率= ($f_{SYS}/32$) /4 3: 时钟频率= ($f_{SYS}/32$) /8 4: 时钟频率= ($f_{SYS}/32$) /16 5: 时钟频率= ($f_{SYS}/32$) /32 6: 时钟频率= ($f_{SYS}/32$) /64 7: 时钟频率= ($f_{SYS}/32$) /128	
1	ADCD1		
2	ADCD2		
3~7	—		保留

ADCD 控制寄存器

S/W Flowchart



软件流程说明

本流程是控制双积分 A/D 实现一次充/放电过程。Timer 0 是充电时做定时用，Timer 1 是放电时计时用。

- 流程开始需先设置好 Timer 的缓存器：给 TMR0 初始值（由需要的定时时间算出），清除为零 TMR1H、TMR1L 计数器缓存器。
- 设置 ADCR 控制缓存器，使 A/D 进入充电模式（ADDISCH1:0=01）A/D 自动开始充电动作。
- 判断比较器的输出脚 ADCMPO 是否为高，高表示 $V_{DSCC} \geq 1/6 V_{DSO}$ ，可以开始定时充电。
- 检测到 ADCMPO 后马上 Enable TMR0，开始定时。
- 检测 TMR0 的溢出旗标 TOF，查询定时充电是否完成。
- TOF=1，则清除 TOF 旗标和清除为零 TMR1H、TMR1L 计数器缓存器。
- 将 ADCR 控制缓存器设置为放电模式（ADDISCH1:0=10），A/D 自动开始放电动作。
- Enable TMR1，开始为放电时间计时。
- 一旦检测到 ADCMPO 由高变低，马上停止 TMR1 计数，并使 ADC 停止工作。

这样就完成一次充电/放电的过程。

软件程序说明

```

INCLUDE HT46R74D-1.INC
INCLUDE MACRO.INC

DATA_MAIN .SECTION 'DATA'
R_HHDB    ?
R_H  DB   ?
R_L  DB   ?
TEMPDB   ?
BUF1DB   ?

CODE_MAIN .SECTION AT 00H 'CODE'

        ORG     00H
        JMP     MLP_Main

        ORG     20H

MLP_MAIN:
        CLR     PAC                ; Initial I/O port
    
```

```

CLR      PBC
CLR      PA
CLR      PB
CLR      INTC0          ; close interrupt
CLR      INTC1
CLR      EADCR
MMOV     CHPRC,63H      ; open charger pump
CALL     DELAY
SET      CHPRC.2
NOP

MMOV     TMR0,00H      ; set charging time
CLR      TMR1L          ; clear TMR1 register
CLR      TMR1H
CLR      TMR1HH

L_LOOP:
MMOV     ADCR,43H      ; open A/D and set to charging mode
SNZ      ADCMPO        ; ADCMPO=1?
JMP      $-1

MMOV     TMR0C,9EH     ; TMR0 set to internal mode and enable
SNZ      T0F           ; T0F=1?
JMP      $-1
CLR      TMR0C.4       ; timeout,close TMR0
CLR      T0F

MMOV     ADCR,45H      ; set A/D to discharging mode and enable
MMOV     TMR1C,98H     ; TMR1 set to internal mode
SZ       ADCMPO        ; ADCMPO=0?
JMP      $-1
CLR      TMR1C.4       ; stop TMR1
CLR      ADPWREN       ; close A/D

MMOV     R_HH,TMR1HH   ; read TMR1 count register
MMOV     R_H,TMR1H
MMOV     R_L,TMR1L
CLR      TMR1L          ; clear TMR1 count register
CLR      TMR1H
CLR      TMR1HH
JMP      L_LOOP

;;-----
DELAY:          ; delay subroutine
MMOV     TEMP,200
MMOV     BUF1,200
SDZ      BUF1
JMP      $-1
CLR      WDT
SDZ      TEMP
JMP      $-6
RET

```

版本记录

版本: V1.10

修改人员: 王赞臣 (深圳分公司)

修改日期: 2007年7月13日

修改内容:

- 修改 IC 外围电路原理图 (图 1)
- 修改 Dual Slope ADC 的内部结构示意图 (图 2)
- 修改 A/D 放大级的放大电路图及其说明 (图 3)